

```

theory Problem-789
  imports Complex-Main
begin

```

Find all functions  $f : \mathbb{R} \rightarrow \mathbb{R}$  satisfying

$$f(f(x) + y) = 4yf(x) + f(x^2 - y).$$

```

theorem
  fixes f :: real => real
  shows (∀ x y. f (f x + y) = 4*y*f x + f (x^2 - y))
    ⟷ (∀ x. f x = x^2) ∨ (∀ x. f x = 0)
  (is (∀ x y. ?eqn x y) ⟷ -)
proof
  assume ∀ x y. ?eqn x y
  then have eqn: ?eqn x y for x y by auto
  have [simp]: f (x^2 + f x) = 4*x^2*f x + f 0 for x
    using eqn[where y=x^2 and x=x]
    unfolding power2-eq-square by smt
  have opts: f x = 0 ∨ f x = x^2 for x
    using eqn[where y=-f x and x=x, simplified]
    by auto
  {
    fix a
    presume f a ≠ a^2
    then have a ≠ 0 and [simp]: f a = 0
      using opts by fastforce+
    fix y
    have *: f y = f (a^2 - y)
      using eqn[where x=a and y=y] by simp
    presume f y ≠ 0
    hence f y = y^2 using opts by auto
    moreover from * and ⟨f y ≠ 0⟩ have f (a^2 - y) = (a^2 - y)^2
      using opts by auto
    ultimately have y^2 = (a^2 - y)^2
      using * by auto
    with ⟨a ≠ 0⟩ have a^2 = 2*y
      by (simp add: power2-eq-square) algebra
  }
  hence **: f a ≠ a^2 ⟹ 2*y ≠ a^2 ⟹ f y = 0 for a y
    by fastforce

  {
    fix a and y
    assume f a ≠ a^2 and 2*y = a^2
    moreover obtain b where 2*y ≠ b^2 and b ≠ 0 and b ≠ y
      by (smt four-x-squared one-power2)
    ultimately have f b ≠ b^2 using ** [where y=b and a=a]
      by simp
    with ** [where a=b and y=y] and ⟨2*y ≠ b^2⟩
    have f y = 0 by simp
  }
  with ** have f a ≠ a^2 ⟹ f y = 0 for a y
    by blast
  thus (∀ x. f x = x^2) ∨ (∀ x. f x = 0)
    using opts by blast
next
  assume (∀ x. f x = x^2) ∨ (∀ x. f x = 0)
  then show ∀ x y. ?eqn x y
    apply (auto simp add: power2-eq-square)
    apply (thin-tac ∀ x. f x = x * x)
    by algebra
qed

```

end